
Variational Calculation of the Hydrogen Molecular Cation (H+
2 ) using Maple

C. W. David
Department of Chemistry
University of Connecticut

Storrs, Connecticut 06269-3060
(Dated: February 28, 2007)

I. SYNOPSIS

It is of interest to carry out a full calculation of the
H+

2 quantum mechanical energy as a function of the in-
ternuclear distance so that one understands that the 2-
electron problem’s difficulties are, at least partially, due
to the electron-electron repulsion.

II. INTRODUCTION

Elementary texts show the ground state wave function
of H+

2 to be

1σ ∼ 1sA + 1sB

where 1si means a 1s orbital located on nucleus i.

FIG. 1: The Elliptical Coordinate System for Diatomic
Molecules. The µ coordinate is not depicted.

Next, the two wave functions in functional[1] rather
than in “symbolic” coordinates:

ψA = e−krA = 1sA;ψB = e−krB = 1sB

where

rA =
√
x2 + y2 + (z −R/2)2

and

rB =
√
x2 + y2 + (z +R/2)2

(R = internuclear distance, with nuclei at
(0,0,±R/2)). Finally, we re-write the wave function in
elliptical coordinates: where

λ =
rA + rB

R

and

µ =
rA − rB

R

giving (in Maple format NOTE, the following is
WRONG. It is being left in for use in a class exercise
in finding and eliminating errors in coding):

#============================================
assume(k>0,R>0);
psi_A := exp(-k*(R/2)*(lambda-mu));
psi_B := exp(-k*(R/2)*(lambda+mu));
#============================================

psi A := e(−1/2 k˜R˜ (λ−µ))

psi B := e(−1/2 k˜R˜ (λ+µ))

The tildes are a result of the “assume(k > 0, R > 0)”
statement, i.e., the assumption to be used that these two
“variables” are to regarded as positive definite. They are
necessary so that future integrations will “work”.

III. THE OVERLAP INTEGRALS

SAA =
∫

all space

ψ2
Adτ

where, in elliptical coordinates,

dτ =
R3

8
dφ
(
λ2 − µ2

)
dλdµ

The term (t1) corresponds to the S(AA) integrand in el-
liptical coordinates.

Typeset by REVTEX

#============================================
t1 := (psi_A**2)*2*Pi*
((R**3)/8)*(lambda**2-mu**2):
#============================================

2*π is the result of dφ integration, and
((R**3)/8)(λ∗∗2−µ∗∗2) dλdµ is the rest of the volume
element which, along with dφ corresponds to dx dy dz.

A. Integration of Mono-Nuclear Overlap (SAA and
SBB)

#============================================
t2 := expand(int(t1,mu=+1..-1)):
#============================================

This line (creating t2) corresponds to integrating over µ
from +1 to -1 (as we know this is the domain), and then
expanding the result,

#============================================
SAA := int(t2,lambda=1..infinity);
#============================================

we finally carry out the second (λ) integral (arriving at
the overlap integral SAA).

SAA :=
π

k˜3

Just as a check, we integrate over the B orbital, and
obtain the same normalization constant.

#============================================
t11 := (psi_B**2)*2*Pi*((R**3)/8)*
(lambda**2-mu**2):
t21 := expand(int(t11,mu=+1..-1)):
t21 := collect(t21,exp(k*R*lambda)):
SBB := int(t21,lambda=1..infinity);
#============================================

and here is the output, showing that SAA = SBB :

SBB :=
π

k˜3

Notice that neither of these integrals is a function of R.

B. Setting Up the A-B Overlap Integral

Now, we attempt the overlap integral itself. Notice the
product of wave functions.

ψAB =
∫

all space

ψAψBdτ

t3 := 2*Pi*psi_A*psi_B*((R**3)/8)*
(lambda**2-mu**2):

This is the famous SAB integral.

#============================================
t4 := expand(int(t3,mu=+1..-1)):
t4 := collect(t4,exp(k*R*lambda)):
SAB := expand(int(t4,lambda=1..infinity)):
SAB := collect(SAB,exp(k*R));
#============================================

SAB :=

1
3
πR˜2

k˜
+
πR˜
k˜2 +

π

k˜3

e(k˜R˜)

Here, we notice that the resultant overlap is a function
of R.

IV. ENERGY OPERATORS

Here we introduce the Lapla-
cian in elliptical coordinates. (See
http://digitalcommons.uconn.edu/chem educ/5)

∇2 =
4

R2 (λ2 − µ2)

{(
∂
(
(λ2 − 1) ∂

∂λ

)
∂λ

)
+∂

(
(1− µ2) ∂

∂µ

)
∂µ

+

∂
(

λ2−µ2

(λ2−1)(1−µ2)
∂

∂φ

)
∂φ


First, we form ∇2ψA:

#============================================
#start HAA section:
t5 := 4/((R**2)*(lambda**2-mu**2))*
(diff((lambda**2-1)*diff(psi_A,lambda),lambda)+
diff((1-mu**2)*diff(psi_A,mu),mu)):
#============================================

After operating with the Laplacian, we attempt the inte-
gration by, first, left multiplying with the A-orbital (later
we will left multiply by the B-orbital), and then integrat-
ing, i.e., ∫

all space

ψA∇2ψAdτ

#============================================
t6 :=psi_A* 2*Pi*((-hbar**2)/(2*m))
*t5*(R**3/8)*(lambda**2-mu**2):
KEAA := int(t6,mu=+1..-1):
KEAA := int(KEAA,lambda=1..infinity):
KEAA := collect(KEAA,exp(k*R));
#============================================

KEAA :=
1
2
π hbar2

k˜m

A. Part of the Exchange Integral

This is now the part where we operate with the Lapla-
cian on the B-orbital, but left multiply by the A-orbital.

KEAB =
∫

all space

ψA∇2ψBdτ

First, we evaluate

∇2ψB

#===========================================
#HAB section
t5B := 4/((R**2)*(lambda**2-mu**2))*
(diff((lambda**2-1)*diff(psi_B,lambda),lambda)+
diff((1-mu**2)*diff(psi_B,mu),mu)):
#===========================================

and then left multiplying by ψA we obtain

#===========================================
t61 :=psi_A* 2*Pi*((-hbar**2)/(2*m))*
t5B*(R**3/8)*(lambda**2-mu**2):
#===========================================

t61 is, along with atomic constants, t5B left multiplied
by psi A.

#===========================================
KEAB := int(t61,mu=+1..-1):
KEAB := collect(KEAB,exp(k*R)):
KEAB := int(KEAB,lambda=1..infinity):
KEAB := collect(KEAB,exp(k*R));
#===========================================

The final kinetic energy term KEAB ,

KEAB := −1
6
e(−k˜R˜)

π hbar2 (−3 + k˜2 R˜2 − 3 k˜R˜)
k˜m

B. The Potential Energy Operator

We need to integrate∫
all space

ψi ∗
ZAe

2

rA
∗ ψjdτ

and ∫
all space

ψi ∗
ZBe

2

rB
∗ ψjdτ

for all i and j, dividing by rA = (R/2) ∗ (λ − µ) will
give one kind of cancellation, while dividing by rB =
(R/2) ∗ (λ + µ) will give the other kind of cancellation.

(λ2 − µ2) = (λ+ µ) ∗ (λ− µ), so when dividing by rA =
(R/2) ∗ (λ− µ) one loses one term of this product i.e.,

R3

8 (λ2 − µ2)
R
2 (λ± µ)

=
R2

4
(λ∓ µ)

while when dividing by the other one loses the other.

#===========================================
I1_a := -2*Pi*Z_A*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda+mu);
#===========================================

I1 a is the Coulomb attraction term (1/r B) concerning
nucleus A

#===========================================
I2_a := -2*Pi*Z_B*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda-mu);
#===========================================

I2 a is the same kind of term concerning nucleus B.

#===========================================
I1_b := int(I1_a,mu=+1..-1):
I2_b := int(I2_a,mu=+1..-1):
I1_c := int(I1_b,lambda=1..infinity):
I2_c := int(I2_b,lambda=1..infinity):
H_AA := KEAA+I1_c+I2_c:
#===========================================

#===========================================
I3_a := -2*Pi*Z_A*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda+mu):
#===========================================

I3 a is the Coulomb attraction term (1/r B) concerning
nucleus A

#===========================================
I4_a :=
-2*Pi*Z_B*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda-mu):
#===========================================

I4 a is the Coulomb attraction term (1/r A) concerning
nucleus B.

#===========================================
I3_b := int(I3_a,mu=+1..-1):
I4_b := int(I4_a,mu=+1..-1):
I3_c := int(I3_b,lambda=1..infinity):
I4_c := int(I4_b,lambda=1..infinity):
#===========================================

We now form the two states, bonding and anti bonding:

#===========================================
H_AB := KEAB + I3_c+I4_c:
Energy_one := (H_AB-H_AA)/(SAB-SAA):
Energy_two := (H_AB + H_AA)/(SAB+SAA):
#===========================================

Here we specialize to H+
2 :

#===========================================
Energy_one := subs(Z_A=1,Z_B=1,Energy_one):
Energy_two := subs(Z_A=1,Z_B=1,Energy_two):
#===========================================

Here we transform to atomic units:

#===========================================
Energy_one :=
subs(esq=1,hbar=1,m=1,Energy_one):
Energy_two := subs(esq=1,hbar=1,m=1,Energy_two);
#===========================================

(Note, please, that the following lines have been edited
to fit on the page properly.)

Energy two(numerator) :=

− 1
6k
e(−kR)π

(
(−3 + k2R2 − 3kR)− 4

kR + 1
k

)
+

π

2k
Denominator =

1
3

(
πR2

k
+
πR
k2

+
π

k3

)
e(−kR) +

π

k3

#===========================================
plot3d(\{Energy_one+1/R,Energy_two+1/R\}
,k=1.5..2.5
,R=0.2..4,axes=BOXED);
#===========================================

#===========================================
En := subs(R=0.5,Energy_two+1/0.5):
plot(En,k=0.0..2.0,labels=[‘k‘,‘Energy‘],
title=‘Variation of Exponential k‘);
#===========================================

V. DISCUSSION OF FIGURES

We suggest (above) how to do the plotting of the re-
sults of these calculations, but, below, display plots which

are not coded. For pedagogical purposes, one can only
stress that involvement with the code itself is the only
way to appreciate what is going on, and just reading this
or that commentary is only partially enlightening. Mas-
tery comes with effort.

The plots shown home in on two facts about these com-
putations. First, the electronic energy of the H+

2 molec-
ular cation is a function of the internuclear distance, and
splitting diagrams which show 1s orbitals splitting into σ
and σ∗ are diagrams at a fixed R value, generally where
the energy of the bonding state is a minimum! But Fig-
ure 3 we can see clearly the splitting between the 1σ and
the 1σ∗ in the vicinity of R =∼ 2au.

Second, it is interesting to note how weak the minimum
is in the LCAO-MO approximation. In fact, Maple itself
could not find the minimum! Hence the approximate
value employed (0.78) for k.

Finally, we note that the entire exercise is wrong. The
limits as R→∞ are positive, while they should be neg-
ative, since at that limit, we have an H atom and a pro-
ton, and the H atom’s electron is in the 1s state! Figure
3 clearly shows that at R > 5 the energies are asymptot-
ically positive!

Further, 4 gives no hint that there is an error here, i.e.,
the seduction of good looking graphs lowers our skepti-
cism concerning accuracy!

After more trouble than I care to discuss, the following
worked:

VI. HERE IS THE CORRECTED VERSION

> restart;
> with(plots):
> assume(k>0,R>0);
> psi_A := exp(-k*(R/2)*(lambda-mu));
> psi_B := exp(-k*(R/2)*(lambda+mu));
> r_A := (R/2)*(lambda-mu);
> r_B := (R/2)*(lambda+mu);
> dtau := ((R^3)/8)*(lambda^2-mu^2);

> #==================================================
> t1 := (psi_A**2)*2*Pi*((R**3)/8)*(lambda**2-mu**2):
> t2 := expand(int(t1,mu=-1..1)):
> t2 := collect(t2,exp(k*R*lambda)):
> print (‘here is SAA‘);
> SAA := int(t2,lambda=1..infinity);
> psi_A := psi_A/sqrt(SAA);#normalize

> t11 := (psi_B**2)*2*Pi*((R**3)/8)*(lambda**2-mu**2):
> t21 := expand(int(t11,mu=-1..1)):
> t21 := collect(t21,exp(k*R*lambda)):
> print (‘here is SBB‘);
> SBB := int(t21,lambda=1..infinity);
> psi_B := psi_B/sqrt(SBB);
> t3 := 2*Pi*psi_A*psi_B*((R**3)/8)*(lambda**2-mu**2):
> t4 := expand(int(t3,mu=-1..1)):
> t4 := collect(t4,exp(k*R*lambda)):
> SAB := expand(int(t4,lambda=1..infinity)):
> SAB := collect(SAB,exp(k*R));
> #start HAA section:
> t5 := 4/((R**2)*(lambda**2-mu**2))*
> (diff((lambda**2-1)*diff(psi_A,lambda),lambda)+
> diff((1-mu**2)*diff(psi_A,mu),mu)):
> t6 :=psi_A* 2*Pi*((-hbar**2)/(2*m))*t5*dtau
> -2*Pi*Z_A*esq*(1/r_A)*psi_A*psi_A*dtau
> -2*Pi*Z_A*esq*(1/r_B)*psi_A*psi_A*dtau:
> HAA := int(t6,mu=-1..1):
> HAA := int(HAA,lambda=1..infinity);
> #HAB section
> t5B := 4/((R**2)*(lambda**2-mu**2))*
> (diff((lambda**2-1)*diff(psi_B,lambda),lambda)+
> diff((1-mu**2)*diff(psi_B,mu),mu)):
> t61 :=psi_A* 2*Pi*((-hbar**2)/(2*m))*t5B*dtau
> -2*Pi*Z_A*esq*(1/r_A)*psi_B*psi_A*dtau
> -2*Pi*Z_A*esq*(1/r_B)*psi_B*psi_A*dtau:
> HAB := int(t61,mu=-1..1):
> HAB := collect(HAB,exp(k*R)):
> HAB := int(HAB,lambda=1..infinity):
> HAB := collect(HAB,exp(k*R));
> Energy_one := (HAA-HAB)/(1-SAB):
> Energy_two := (HAB + HAA)/(SAB+1):
> #specialize to homonuclear Z=1 case
> Energy_one := subs(Z_A=1,Z_B=1,Energy_one);
> Energy_two := subs(Z_A=1,Z_B=1,Energy_two);
> Energy_one := subs(esq=1,hbar=1,m=1,Energy_one);
> Energy_two := subs(esq=1,hbar=1,m=1,Energy_two);

Energy one :=
−1

6
e(−k˜R˜) π (−3 + k˜2 R˜2 − 3 k˜R˜)

k˜
− 1

2
π

k˜
1
3
πR˜2

k˜
+
πR˜
k˜2 +

π

k˜3

e(k˜R˜)
− π

k˜3

> plot3d({Energy_one+1/R,Energy_two+1/R
> },k=1.0..1.5,R=0.2..4,axes=BOXED);

> En5 := subs(R=1.0,Energy_two+1/R):
> En6 := subs(R=1.2,Energy_two+1/R):
> En7 := subs(R=1.4,Energy_two+1/R):
> En8 := subs(R=1.6,Energy_two+1/R):
> En9 := subs(R=1.8,Energy_two+1/R):
> En10 := subs(R=2.0,Energy_two+1/R):
> En11 := subs(R=2.2,Energy_two+1/R):
> En12 := subs(R=2.4,Energy_two+1/R):
> En13 := subs(R=2.6,Energy_two+1/R):
> En14 := subs(R=2.8,Energy_two+1/R):
> En15 := subs(R=3.0,Energy_two+1/R):
> plot({En5,En6,En7,En8,En9,En10,En11,En12,En13,En14,En15
> },k=0.0..2.0,labels=[‘k‘,‘Energy‘],title=‘Variation of Exponential
> k‘);
> contourplot(Energy_two+1/R,R=0.8..3.5,k=0.8..2.0,contours = 80);
> #minimize(Energy_two+1/R,R=0.8..3.5,k=0.8..2.0,location);#FAILED ON
> TIME
> #choose 1.2 (by eye);
> plot({subs(k=1.2,Energy_two+1/R),
> subs(k=0.78,Energy_one+1/R)
> },R=0.1..5.0,labels=[‘R‘,‘Energy‘],title=‘E(R)‘);
> limit1 := limit(subs(k=1.2,Energy_one+1/R),R=infinity);
> limit2 := limit(subs(k=1.2,Energy_two+1/R),R=infinity);

Warning, the name changecoords has been redefined

psi A := e(−
k˜ R˜ (λ−µ)

2 )

psi B := e(−
k˜ R˜ (λ+µ)

2 )

r A :=
R˜ (λ− µ)

2

r B :=
R˜ (λ+ µ)

2

dtau :=
R˜3 (λ2 − µ2)

8
here is SAA

SAA :=
π

k˜3

psi A :=
e(−

k˜ R˜ (λ−µ)
2 ) k˜√
π

k˜
here is SBB

SBB :=
π

k˜3

psi B :=
e(−

k˜ R˜ (λ+µ)
2 ) k˜√
π

k˜

SAB :=

1
3

k˜2 R˜2 + k˜R˜ + 1

e(k˜R˜)

HAA := −1
2
(2Z Aesq m (e(k˜R˜))2 R˜ k˜− 2Z Aesq m− k˜2 hbar2 R˜ (e(k˜R˜))2

− 2Z Aesq m k˜R˜ + 2 (e(k˜R˜))2 Z Aesq m)
/

(R˜ (e(k˜R˜))2m)

HAB := −1
6
k˜(hbar2 k˜3 R˜2 − 3 hbar2 k˜ + 12Z Aesq m− 3 hbar2 k˜2 R˜

+ 12Z Aesq m k˜R˜)e(−k˜R˜)/m

Energy two := (−1
6

k˜ (hbar2 k˜3 R˜2 − 3 hbar2 k˜ + 12 esq m− 3 hbar2 k˜2 R˜ + 12 esq m k˜R˜) e(−k˜R˜)

m

− 1
2
(2 esq m (e(k˜R˜))2 R˜ k˜− 2 esq m− k˜2 hbar2 R˜ (e(k˜R˜))2 − 2 esq m k˜R˜

+ 2 (e(k˜R˜))2 esq m)
/

(R˜ (e(k˜R˜))2m))
/1

3
k˜2 R˜2 + k˜R˜ + 1

e(k˜R˜)
+ 1


Energy one := (−1

2
2 (e(k˜R˜))2 R˜ k˜− 2− k˜2 R˜ (e(k˜R˜))2 − 2 k˜R˜ + 2 (e(k˜R˜))2

R˜ (e(k˜R˜))2

+
1
6

k˜ (k˜3 R˜2 − 3 k˜ + 12− 3 k˜2 R˜ + 12 k˜R˜) e(−k˜R˜))
/(

1−

1
3

k˜2 R˜2 + k˜R˜ + 1

e(k˜R˜)

)

Energy two := (−1
6

k˜ (k˜3 R˜2 − 3 k˜ + 12− 3 k˜2 R˜ + 12 k˜R˜) e(−k˜R˜)

− 1
2

2 (e(k˜R˜))2 R˜ k˜− 2− k˜2 R˜ (e(k˜R˜))2 − 2 k˜R˜ + 2 (e(k˜R˜))2

R˜ (e(k˜R˜))2
)
/(

1
3

k˜2 R˜2 + k˜R˜ + 1

e(k˜R˜)
+ 1

)

VII. WRONG FIGURES

FIG. 2: WRONG Energies of the ground and first excited states of the electron of the H+
2 cation

limit1 := −0.4800000000
limit2 := −0.4800000000

[1] C. W. David, When are LCAO-MOs not Eigenfunctions?,
J. Chem. Ed., 59, 288(1982).



9

E(R)

0

2

4

6

8

10

Energy

1 2 3 4 5

R~

FIG. 3: WRONG Energies of the ground and first excited states of the electron of the H+
2 cation assuming a fixed value of

k=0.78, plotted as a function or the internuclear distance R

FIG. 4: WRONG Contours of the Energy as a function of R and k

FIG. 5: Energy as a function of R and k (σ and σ∗)

FIG. 6: Energy as a function of k (σ)

FIG. 7: Energy contour map as a function of R and k (σ)

FIG. 8: Energy as a function of R and k (σ and σ∗)

